Paper 204–29

# An Introduction to ODS for Statistical Graphics in SAS® 9.1

Robert N. Rodriguez
SAS Institute Inc.
Cary, North Carolina, USA

## ABSTRACT

In SAS 9.1, over two dozen SAS/STAT® and SAS/ETS® procedures have been modified to use an experimental extension to the Output Delivery System (ODS) that enables them to create statistical graphics as automatically as tables are currently created. This extension, referred to as *ODS for Statistical Graphics* (*ODS Graphics for short*), provides commonly used displays, including scatter plots, histograms, box-and-whisker plots, and contour plots, in ODS output. Many ODS features for tables, such as destination statements, apply equally to graphics. ODS styles control the appearance and consistency of all graphs, whereas ODS templates control the layout and details of individual graphs. This paper introduces statistical users and other SAS programmers to ODS Graphics. Examples illustrate basic functionality, which requires minimal syntax, and typical graph management tasks. Familiarity with ODS for tables is assumed.

## INTRODUCTION

Graphics are indispensable for modern data analysis, statistical modeling, and data mining. They enrich the analysis by revealing patterns, identifying differences, and expressing uncertainty that would not be readily apparent in tabular output. Effective graphics add visual clarity and rich content to an analytical presentation, and they provoke questions that would not otherwise be raised, stimulating deeper investigation.

In SAS 9.1, a number of SAS/STAT procedures have been modified to use an experimental extension to the Output Delivery System (ODS) that enables them to create statistical graphics as automatically as tables. This facility is referred to as *ODS Statistical Graphics* (or *ODS Graphics* for short), and it is invoked when you provide the experimental ODS GRAPHICS statement prior to your procedure statements. Any procedures that use ODS Graphics then create graphics, either by default or when you specify procedure options for requesting specific graphs.

For example, consider the use of the LIFETEST procedure to create a plot of two estimated survival curves, one for patients treated with a new drug and the second for patients treated with a placebo. Traditionally, this requires three steps: creating output data sets with values of the survival curves and metadata (such as the numbers of patients); modifying the data sets programmatically; and finally using graphics procedures and the annotate facility to create the plot.

With ODS Graphics in SAS 9.1, you can create this display in one short step.

```
ods html body="lifetest.htm";
ods graphics on;

proc lifetest data=drug;
   time surv*censor(1);
   survival plots=(survival hwb);
   strata treatment;
   id patient;
run;

ods html close;
ods graphics off;
```

The ODS HTML statement specifies an HTML destination for the output. The ODS GRAPHICS statement requests ODS Graphics in addition to the usual tabular output, and the PLOTS= option requests the survival plot. The ODS GRAPHICS OFF statement disables ODS Graphics, and the ODS HTML CLOSE statement closes the HTML destination. The survival plot, requested with the SURVIVAL option, is shown in Figure 1. A second plot, requested with the HWB option, is shown in Figure 9.




**Figure 1.** Survival Plot Produced by LIFETEST Procedure

With ODS Graphics, a procedure creates the graphs that are most *commonly* needed for a particular analysis–either by default or with options as in the preceding example. ODS Graphics takes advantage of the analytical context available within statistical procedures, so that the user does not need to reconstruct this content with subsequent programming steps in order to create graphs. In many cases, ODS graphs are automatically enhanced with useful statistical information or metadata, such as sample sizes and $p$-values, which are displayed in an inset box.

The procedures that use ODS Graphics in SAS 9.1 are listed in Appendix A.

**Note:** Statistical graphics created with ODS are experimental in SAS 9.1. Both their appearance and their syntax are subject to change in a future release. The number of procedures and graph types supported with ODS Graphics will grow in future releases.

#### COMPARISON WITH ODS FOR TABLES

In many ways, creating graphics with ODS is analogous to creating tables with ODS. You use

- procedure options and defaults to specify *which graphs are created*
- ODS destination statements (such as ODS HTML) to specify *where graphs are displayed*

Additionally, you can use

- graph names in ODS SELECT and EXCLUDE statements to select and exclude graphs from your output
- ODS styles to control the general appearance and consistency of *all graphs*
- ODS templates to control the layout and details of *individual graphs*. A default template is provided by SAS for each graph.

Graphs and tables are integrated in ODS output, as shown in Figure 2.

In SAS 9.1, the ODS destinations that support ODS Graphics include HTML, LATEX, PRINTER, and RTF; see page 23. Note that the LISTING destination is not supported.

### CREATING GRAPHICS WITH STYLE

ODS styles control the overall look of your output. A style definition provides formatting information for specific visual aspects of your SAS output. Starting with SAS 9, ODS styles also include graphical appearance information such as line and marker properties in addition to font and color information. In SAS 9.1, ODS styles provide the defaults for common elements of statistical graphics created with ODS Graphics, including fitted lines, confidence and prediction bands, and outliers.

You can specify a style using the STYLE= option in a valid ODS destination, such as HTML, PDF, RTF, or PRINTER. Each style produces output with the same content, but a somewhat different visual appearance. Of the SAS-supplied styles for SAS 9.1, four are recommended for use with ODS Graphics:

- Analysis
- Default
- Journal
- Statistical

Figure 2 and Figure 3 illustrate the difference between the Default and the Journal styles for the HTML destination. Note that the appearance of tables and graphics is coordinated by a style.




**Figure 2.** HTML Output with Default Style




**Figure 3.** HTML Output with Journal Style

## SAVING YOUR GRAPHS

Both tables and graphs are saved in the ODS output file produced for a destination. However, individual graphs can also be saved in files, which are produced in a specific graphics image file type, such as GIF or PostScript. This enables you to access individual graphs for inclusion in a document.

For example, you can save graphs in PostScript files to include in a paper that you are writing with LaTeX. Likewise, you can save graphs in GIF files to include in an HTML document. Graphics image files types are discussed further on page 23.

## HOW TO GET STARTED WITH ODS GRAPHICS

If you are trying out ODS Graphics for the first time, begin by reading the graphics examples in the chapters of the SAS/STAT and SAS/ETS user's guides for procedures that use ODS Graphics in SAS 9.1.

To take full advantage of ODS Graphics, you will need to learn more about ODS destinations, output files, and image file types for graphics, as well as ways to access and include individual graphs in reports and presentations. This is explained in Chapter 15 of the *SAS/STAT 9.1 User's Guide*, "Statistical Graphics Using ODS (Experimental)". All of the topics discussed in this paper are covered in greater detail in Chapter 15. A similar chapter is provided in the *SAS/ETS 9.1 User's Guide*.

## GALLERY OF GRAPH TYPES

This section provides examples of ODS graphs produced with SAS/STAT and SAS/ETS procedures in SAS 9.1. All of these examples were created using the Default style and the HTML destination.




**Figure 4.** GAM Procedure: Smoothed Components for Generalized Additive Model




**Figure 5.** GLM Procedure: Analysis of Covariance




**Figure 6.** HPF Procedure: Predicted Values from Forecasting Model




**Figure 7.** KDE Procedure: Bivariate Kernel Density Estimate




**Figure 8.** KDE Procedure: Bivariate Histogram and Kernel Density Estimate




**Figure 9.** LIFETEST Procedure: Hall-Wellner Confidence Bands for Survival Functions




**Figure 10.** LOESS Procedure: LOESS Smooth for Scatter Plot




**Figure 11.** LOGISTIC Procedure: Estimated Probability for Logistic Regression




**Figure 12.** MIXED Procedure: Influence Diagnostics for Fixed Effects in Mixed Model




**Figure 13.** MIXED Procedure: Studentized Residuals for Mixed Model




**Figure 14.** MIXED Procedure: Box Plot for Fixed Main Effects in Mixed Model




**Figure 15.** PHREG Procedure: Cumulative Residuals for Proportional Hazards Model




**Figure 16.** PLS Procedure: Score Plots for Partial Least Squares Analysis




**Figure 17.** PRINCOMP Procedure: Principal Component Scores




**Figure 18.** REG Procedure: Regression Diagnostics




**Figure 19.** REG Procedure: Simple Linear Regression




**Figure 20.** TIMESERIES Procedure: Time Series Decomposition Plots




**Figure 21.** UCM Procedure: Prediction Errors for Unobserved Component Model




**Figure 22.** UCM Procedure: Prediction Error Autocorrelation Plot for Unobserved Component Model

## MANAGING YOUR ODS GRAPHICS

The following examples illustrate various aspects of managing graphs created with ODS Graphics.

### EXAMPLE 1: SELECTING AND EXCLUDING GRAPHS

This example illustrates how to select and exclude ODS graphs from your output.

By default, the REG procedure produces a panel, shown in Figure 18, which contains six different diagnostics plots. To display all the plots individually, you can specify the option PLOTS(UNPACK) in the PROC REG statement. However, if you are interested in displaying only one of these plots, say the Cook's D plot, you can use the ODS SELECT statement to make the selection.

In order to specify a particular graph with the ODS SELECT statement you must know it name. There are three ways to determine a graph name*. First, you can look it up in the Details section titled "ODS Graph Names" in the procedure chapter in the *SAS/STAT 9.1 User's Guide*. Second, you can use the Results window to view the names of ODS graphs created in your SAS session. Third, you can run the procedure with the ODS TRACE ON statement, which requests a record of the output objects created by ODS, as illustrated by the following statements.

```
ods trace on;

ods html style = Journal;
ods graphics on;

proc reg data = Class plots(unpack);
   model Weight = Height;
```

*The ODS graph name is not the same as the name of the file containing the graph. File names are discussed on page 24 of this paper.

```
run;
quit;

ods graphics off;
ods html close;

ods trace off;
```

The record is displayed in the SAS log as partially shown in Figure 23.

```
Output Added:
-------------
Name:        CooksD
Label:       Cook's D
Template:    Stat.REG.Graphics.CooksD
Path:        Reg.MODEL1.ObswiseStats.Weight.DiagnosticPlots.CooksD
-------------
```

**Figure 23.** Partial ODS TRACE Record in SAS Log

The following statements restrict the output to the Cook's D plot, which is shown in Figure 24.

```
ods html style = Journal;
ods graphics on;

ods select CooksD;

proc reg data = Class plots(unpack);
   model Weight = Height;
run;
quit;

ods graphics off;
ods html close;
```




**Figure 24.** Cook's D Plot Created with Journal Style

Conversely, you can use the ODS EXCLUDE statement to display all the output with the exception of a particular subset of tables or graphs.

#### EXAMPLE 2: SPECIFYING THE STYLE OF A GRAPH

ODS style definitions include a number of graph elements that correspond to general features of statistical graphics, such as titles and fitted lines. The attributes of these elements, such as fonts and colors, provide the defaults for options in graph templates provided by SAS. Consequently, you can change all of your graphs in a consistent manner by simply selecting a different style. For example, by specifying the Journal style, you can create gray-scale graphs and tables that are suitable for statistical journals, reports, and other publications that require black-and-white figures.

Styles are specified with the STYLE= option in an ODS destination statement as illustrated in Figure 24 and in Example 5.

#### EXAMPLE 3: CREATING GRAPHS WITH TOOL TIPS IN HTML

This example demonstrates how to request graphics in HTML with tool tip displays, which appear when you move a mouse over certain features of the graph. When you specify the HTML destination and the IMAGEFMT=STATICMAP option in the ODS GRAPHICS statement, then the HTML file output file is generated with an image map of coordinates for tool tips. The individual graphs are saved as GIF files.

The following statements fit a mixed model with random intercepts and slopes for each child in a data set with repeated growth measurements for 27 children. The experimental BOXPLOT option in the PROC MIXED statement requests box plots of observed values and residuals for each classification main effect in the model (Gender and Person).

```
ods html;
ods graphics on / imagefmt = staticmap;

proc mixed data=pr method=ml boxplot(npanel=15);
   class Person Gender;
   model y = Gender Age Gender*Age;
   random intercept Age / type=un subject=Person;
run;

ods graphics off;
ods html close;
```

The NPANEL=15 suboption limits the number of box plots per graph to at most 15. Here, the conditional residuals of the Person effect are displayed in two graphs consisting of 15 and 12 boxes, respectively. Figure 25 displays the second of these two graphs.

Moving the mouse over a box plot displays a tool tip with summary statistics for the corresponding child.




**Figure 25.** Box Plot with Tool Tips

**Note:** Graphics with tool tips are only supported for the HTML destination.

### EXAMPLE 4: CREATING GRAPHS FOR A PRESENTATION

The RTF destination provides a convenient way to create ODS graphs for inclusion in a document or presentation. You can specify the ODS RTF statement to create a file that is easily imported into a word processor (such as Microsoft Word or WordPerfect) or a presentation (such as Microsoft PowerPoint).

The following statements request a loess fit and save the output in the file loess.rtf.

```
ods rtf file = "loess.rtf";
ods graphics on;

proc loess data = one;
   model y = x / clm residual;
run;

ods graphics off;
ods rtf close;
```

The output file includes various tables and the following plots: a plot of selection criterion versus smoothing parameter, a fit plot with 95% confidence bands, a plot of residual by regressors, and a diagnostics panel. The fit plot is shown in Figure 26.




**Figure 26.** Fit Plot in RTF Output

If you are running SAS in the Microsoft Windows operating system, you can open the RTF file in Microsoft Word and simply copy and paste the graphs into Microsoft PowerPoint. In general, RTF output is convenient for exchange of graphical results between Windows applications through the clipboard.

Another convenient way to create ODS graphics for a presentation is to use the HTML destination. By default, the individual graphs are created as GIF files, and you can cut and paste them from your HTML output into a Microsoft PowerPoint presentation.

See page 25 for information on how image files are named and saved in a directory.

#### EXAMPLE 5: CREATING GRAPHS IN POSTSCRIPT FILES

This example illustrates how to create individual graphs in PostScript files, which is useful when you want to include them in a LaTeX document.

The following statements specify a LATEX destination with the Journal style, and request a histogram of standardized robust residuals computed with the ROBUSTREG procedure.

```
ods latex style = Journal;
ods graphics on;

proc robustreg plot=reshistogram data=stack;
   model y = x1 x2 x3;
run;
```

```
ods graphics off;
ods latex close;
```

The Journal style displays gray-scale graphs that are suitable for a journal. When you specify the ODS LATEX destination, ODS creates a PostScript file for each individual graph in addition to a LaTeX source file that includes the tabular output and references to the PostScript files. By default these files are saved in the SAS current folder. If you run this example at the beginning of your SAS session, the histogram shown in Figure 27 is saved by default in a file named ResidualHistogram0.ps. See page 24 for details about how graphics image files are named.




**Figure 27.** Histogram Using Journal Style

If you are writing a paper in LaTeX, you can include the graphs in your LaTeX source file by referencing the names of the individual PostScript graphics files. In this situation, you may not need to use the LaTeX source file created by SAS.

If you specify PATH= and GPATH= options in the ODS LATEX statement, your tabular output is saved as a LaTeX source file in the directory specified with the PATH= option, and your graphs are saved as PostScript files in the directory specified with the GPATH= option. This is illustrated by the following statements:

```
ods latex path  = "C:\temp"
          gpath = "C:\temp\ps" (url="ps/")
          style = Journal;
ods graphics on;

   ...SAS statements...

ods graphics off;
ods latex close;
```

The URL= suboption is specified in the GPATH= option to create relative paths for graphs referenced in the LaTeX source file created by SAS.

### EXAMPLE 6: CUSTOMIZING GRAPH TITLES AND AXES LABELS

This example shows how to customize the appearance and content of an ODS graph by modifying its template.

The following statements request a Q-Q plot for robust residuals using PROC ROBUSTREG.

```
ods trace on;
ods html;
ods graphics on;

ods select ResidualQQPlot;

proc robustreg plot=resqqplot data=stack;
   model y = x1 x2 x3;
run;

ods graphics off;
ods html close;
ods trace off;
```

The Q-Q plot is shown in Figure 28.




**Figure 28.** Default Q-Q Plot from PROC ROBUSTREG

The ODS TRACE ON statement requests a record of all the ODS output objects created by PROC ROBUSTREG. A partial listing of the trace record, which is displayed in the SAS log, is shown in Figure 29.

```
Output Added:
-------------
Name:        ResidualQQPlot
Label:       ResidualQQPlot
Template:    Stat.Robustreg.Graphics.ResidualQQPlot
Path:        Robustreg.Graphics.ResidualQQPlot
-------------
```

**Figure 29.** Partial Trace Record for Q-Q Plot

ODS Graphics creates the Q-Q plot from an ODS output data object named "ResidualQQPlot" and a graph template named "Stat.Robustreg.Graphics.ResidualQQPlot," which is the default template provided by SAS.

To display the default template definition, open the Templates window by typing **odstemplates** (or **odst** for short) in the command line. Expand **Sashelp.Tmplmst** and click on the **Stat** folder, as shown in Figure 30.




**Figure 30.** The Templates Window

Next, open the **Robustreg** folder and then open the **Graphics** folder. Then right-click on the "ResidualQQPlot" template icon and select **Edit**. Selecting **Edit** opens a Template Editor window, as shown in Figure 31, which you can use to edit the template.

Graph template definitions are written in an experimental graph template language, which has been added to the TEMPLATE procedure in SAS 9.1. See Appendix C for more information.

In the template, the default title of the Q-Q plot is specified by the two ENTRYTITLE statements. Here _DEPLABEL is a dynamic variable that provides the name of the dependent variable in the regression analysis (the name happens to be y in Figure 28). The default label for the y-axis is specified by the LABEL= suboption of the YAXISOPTS= option for the LAYOUT OVERLAY statement.




**Figure 31.** Default Template Definition for Q-Q Plot

Suppose you want to change the default title to My Residual Analysis for y, and you want to change the y-axis label to Std Robust Residual (M Estimation) to reflect the robust method used. First, replace the two ENTRYTITLE statements with the following statements:

```
ENTRYTITLE "My Residual Analysis for " / padbottom = 5;
ENTRYTITLE _DEPLABEL / padbottom = 5;
```

Next, replace the LABEL= suboption with the following:

```
label = "Std Robust Residual (M Estimation)"
```

Note that you can reuse dynamic text variables such as _DEPLABEL in any text element.

You can then submit the modified template definition as you would any SAS program. You should see the following message in the SAS log:

```
NOTE: STATGRAPH 'Stat.Robustreg.Graphics.ResidualQQPlot' has been
  saved to: SASUSER.TEMPLAT
```

Finally, resubmit the PROC ROBUSTREG statements* on page 20 to display the Q-Q plot created with your modified template, as shown in Figure 32.

*In fact, you do not need to rerun the procedure after you modify a graph template. Instead, you can use the DOCUMENT procedure to replay the graph with the modified template.




**Figure 32.** Q-Q Plot with Modified Title and Y-Axis Label

The modified template "ResidualQQPlot" is used automatically because SASUSER.TEMPLAT occurs before SASHELP.TMPLMST in the ODS search path.

### GRAPHICS IMAGE FILES

Accessing your graphs as individual image files is useful when you want to include them in various types of documents. The default image file type depends on the ODS destination, but there are other supported image file types that you can specify. You can also specify the names for your graphics image files and the directory in which you want to save them.

If you are using an HTML or a LATEX destination, your graphs are individually produced in a specific image file type, such as GIF or PostScript.

If you are using a destination in the PRINTER family or the RTF destination, the graphs are contained in the ODS output file and cannot be accessed as individual image files. However, you can open an RTF output file in Microsoft Word and then copy and paste the graphs into another document, such as a Microsoft PowerPoint presentation; this is illustrated in Example 4 beginning on page 17.

The following table shows the various ODS destinations supported by ODS Graphics, the viewer that is appropriate for displaying graphs in each destination, and the image file types supported for each destination. Note that in SAS 9.1 the LISTING destination does not support ODS Graphics. You must specify a supported ODS destination in order to produce ODS Graphics, as illustrated by all the examples in this chapter.

| Destination | Destination Family | Viewer | Image File Types |
|---|---|---|---|
| DOCUMENT | | Not Applicable | Not Applicable |
| HTML | MARKUP | Browser | GIF (default), JPEG, PNG |
| LATEX | MARKUP | Ghostview | PostScript (default), EPSI, GIF, JPEG, PNG |
| PCL | PRINTER | Ghostview | Contained in PostScript file |
| PDF | PRINTER | Acrobat | Contained in PDF file |
| PS | PRINTER | Ghostview | Contained in PostScript file |
| RTF | | Microsoft Word | Contained in RTF file |

#### NAMING GRAPHICS IMAGE FILES

The names of graphics image files are determined by a *base file name*, an *index counter*, and an *extension*. By default, the base file name is the ODS graph name (graph names are discussed in Example 1). The counter is set to zero when you begin a SAS session, and it is increased by one after you create a graph, independently of the graph type or the procedure that creates it. The extension indicates the image file type.

For instance, suppose you run the following statements at the beginning of a SAS session.

```
ods html;
ods graphics on;

proc kde data = bivnormal;
   bivar x y / plots = contour surface;
run;

ods graphics off;
ods html close;
```

The two graphics image files created are Contour0.gif and SurfacePlot1.gif, which correspond to Figure 7 and Figure 8. If you immediately rerun this example, then ODS creates the same graphs in different image files named Contour2.gif and SurfacePlot3.gif. You can specify the RESET option in the ODS GRAPHICS statement to reset the index counter to zero. This avoids duplication of graphics image files if you are rerunning a SAS program in the same session.

You can specify a base file name for all your graphics image files with the IMAGENAME= option in the ODS GRAPHICS statement. For example:

```
ods graphics on / imagename = "MyName";
```

You can also specify

```
ods graphics on / imagename = "MyName" reset;
```

With the preceding statement, the graphics image files are named MyName0, MyName1, and so on.

You can specify the image file type for the HTML or LATEX destinations with the IMAGEFMT= option in the ODS GRAPHICS statement as follows.

```
ods graphics on / imagefmt = png;
```

### SAVING GRAPHICS IMAGE FILES

Knowing where your graphics image files are saved and how they are named is particularly important if you are running in batch mode or if you plan to access the files for inclusion in a document. The following discussion assumes you are running SAS under the Windows operating system.

Your graphics image files are saved by default in the SAS current folder. If you are using the SAS windowing environment, the current folder is displayed in the status line at the bottom of the main SAS window. If you are running your SAS programs in batch mode, the graphs are saved by default in the same directory where you started your SAS session.

With the HTML and the LATEX destinations, you can specify a directory for saving your graphics image files. With the PRINTER and RTF destinations, you can only specify a directory for your output file.

If you are using the HTML destination, the individual graphs are created as GIF files by default. You can use the PATH= and GPATH= options in the ODS HTML statement to specify the directory where your HTML and graphics files are saved, respectively. This also gives you more control over your graphs. For example, if you want to save your HTML file named test.htm in the C:\myfiles directory, but you want to save your graphics image files in C:\myfiles\gif, then you specify

```
ods html path  = "C:\myfiles"
         gpath = "C:\myfiles\gif"
         file  = "test.htm";
```

For more information, refer to Chapter 15 of the *SAS/STAT 9.1 User's Guide*.

### APPENDIX A: PROCEDURES SUPPORTING ODS GRAPHICS

The following procedures support ODS Graphics in SAS 9.1:

**Base SAS**

- CORR

**SAS/ETS**

- ARIMA
- AUTOREG
- ENTROPY
- EXPAND
- MODEL
- SYSLIN
- TIMESERIES
- UCM
- VARMAX
- X12

**SAS High-Performance Forecasting**

- HPF

**SAS/STAT**

- ANOVA
- CORRESP
- GAM
- GENMOD
- GLM
- KDE
- LIFETEST
- LOESS
- LOGISTIC
- MI
- MIXED
- PHREG
- PRINCOMP
- PRINQUAL
- REG
- ROBUSTREG

### APPENDIX B: RELATIONSHIP WITH TRADITIONAL HIGH-RESOLUTION GRAPHICS

ODS Graphics are produced completely independently of both line printer plots and traditional high-resolution graphics requested with SAS/GRAPH procedures or with some analysis procedures such as UNIVARIATE and REG. Traditional high-resolution graphics are saved in graphics catalogs and controlled by the GOPTIONS statement. In contrast, ODS Graphics are produced in ODS output (not graphics catalogs) and their appearance and layout are controlled by ODS styles and templates. In SAS 9.1 both line printer plots and traditional high-resolution graphics supported by procedures such as REG continue to be available and are unaffected by the ODS GRAPHICS statement.

### APPENDIX C: THE GRAPH TEMPLATE LANGUAGE

Graph template definitions are written in the *graph template language*, which has been added to the TEMPLATE procedure in SAS 9.1. This language is used to write the default templates for ODS Graphics, which are supplied by SAS. In common applications of the procedures it should not be necessary for the user to modify graph templates, just as it is typically not necessary for the user to modify ODS table templates. The long-term goal of ODS Graphics is to make it possible for procedures to produce graphics as *automatically* as tables in ODS output, and users should rarely need to interact with templates.

**Note:** In SAS 9.1 the graph template language is experimental, as are all other aspects of ODS Graphics, and the syntax is expected to change in a future release of SAS. You can use the language to modify graph templates as illustrated in Example 6, but it is *not* designed as a general facility for annotation or for composing novel graphical displays.

The graph template language includes statements for specifying plot layouts (such as grids or overlays), plot types (such as scatter plots and histograms), and text elements (such as titles, footnotes, and insets). It also provides support for built-in computations (such as histogram binning) and evaluation of expressions. Options are available for specifying colors, marker symbols, and other attributes of plot features.

Graph template definitions begin with a DEFINE STATGRAPH statement in PROC TEMPLATE, and they end with an END statement. The statements available in the graph template language can be classified as follows:

- **Control statements**, which specify conditional or iterative flow of control. By default, flow of control is sequential. In other words, each statement is used in the order in which it appears.
- **Layout statements**, which specify the arrangement of the components of the graph. Layout statements are arranged in blocks that begin with a LAYOUT statement and end with an ENDLAYOUT statement. The blocks can be nested. Within a layout block, you can specify plot, text, and other statement types to define one or more graph components. Statement options provide control for attributes of layouts and components.
- **Plot statements**, which specify a number of commonly used displays, including scatter plots, histograms, contour plots, surface plots, and box plots. Plot statements are always provided within a layout block. The plot statements include options to specify which data columns from the source objects are used in the graph. For example, in the SCATTERPLOT statement used to define a scatter plot, there are mandatory X= and Y= options that specify which data columns are used for the $x$- and $y$-variables in the plot, and there is a GROUP= option that specifies a data column as an optional classification variable.
- **Text statements**, which specify descriptions accompanying the graphs. An entry is any textual description, including titles, footnotes, and legends, and it can include symbols to identify graph elements.

As an illustration, the following statements display the template definition of the scatter plot available with the KDE procedure.

```
proc template;
   define statgraph Stat.KDE.Graphics.ScatterPlot;
      dynamic _TITLE _DEPLABEL _DEPLABEL2;
      layout Gridded;
         layout overlay / padbottom = 5;
            entrytitle _TITLE;
         endlayout;
         scatterplot x=X y=Y /
            markersymbol = GraphDataDefault:markersymbol
            markercolor  = GraphDataDefault:contrastcolor
            markersize   = GraphDataDefault:markersize;
      EndLayout;
   end;
run;
```

The DEFINE STATGRAPH statement in PROC TEMPLATE creates the graph template definition. The DYNAMIC statement defines three dynamic variables. The variable _TITLE provides the title of the graph. The variables _DEPLABEL and _DEPLABEL2 contain the names of the $X$ and $Y$ variables, respectively. You can use these dynamic text variables in any text element of the graph definition.

The overall display is specified with the LAYOUT GRIDDED statement. The title of the graph is specified with the ENTRYTITLE statement inside a layout overlay block, which is nested within the main layout. The main plot is a scatter plot specified with the SCATTERPLOT statement. The options in the SCATTERPLOT statement, which are given after the slash, specify the symbol, color, and size for the markers using indirect references to style attributes of the form `style-element:attribute`. The values of these attributes are specified in the definition of the style you are using, and so they are automatically set to different values if you specify a different style.

The second ENDLAYOUT statement ends the main layout block and the END statement ends the graph template definition.

For details concerning the syntax of the graph template language, refer to the "TEMPLATE Procedure: Creating ODS Statistical Graphics Output (Experimental)" which is available at http://support.sas.com/documentation/onlinedoc/base/.

**ACKNOWLEDGMENTS**

I am grateful to Francisco Chamu, Virginia Clark, and Michael Crotty of SAS Institute for valuable assistance in the preparation of this paper.

**CONTACT INFORMATION**

Robert N. Rodriguez
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
(919) 531-7650
Bob.Rodriguez@sas.com